

NAG Library Routine Document

F04BCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F04BCF computes the solution to a real system of linear equations $AX = B$, where A is an n by n tridiagonal matrix and X and B are n by r matrices. An estimate of the condition number of A and an error bound for the computed solution are also returned.

2 Specification

```
SUBROUTINE F04BCF (N, NRHS, DL, D, DU, DU2, IPIV, B, LDB, RCOND, ERRBND, &
                  IFAIL)
INTEGER          N, NRHS, IPIV(N), LDB, IFAIL
REAL (KIND=nag_wp) DL(*), D(*), DU(*), DU2(N-2), B(LDB,*), RCOND, ERRBND
```

3 Description

The LU decomposition with partial pivoting and row interchanges is used to factor A as $A = PLU$, where P is a permutation matrix, L is unit lower triangular with at most one nonzero subdiagonal element, and U is an upper triangular band matrix with two superdiagonals. The factored form of A is then used to solve the system of equations $AX = B$.

Note that the equations $A^T X = B$ may be solved by interchanging the order of the arguments DU and DL .

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Parameters

- 1: N – INTEGER *Input*
On entry: the number of linear equations n , i.e., the order of the matrix A .
Constraint: $N \geq 0$.
- 2: NRHS – INTEGER *Input*
On entry: the number of right-hand sides r , i.e., the number of columns of the matrix B .
Constraint: $NRHS \geq 0$.
- 3: DL(*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array DL must be at least $\max(1, N - 1)$.
On entry: must contain the $(n - 1)$ subdiagonal elements of the matrix A .
On exit: if $IFAIL \geq 0$, DL is overwritten by the $(n - 1)$ multipliers that define the matrix L from the LU factorization of A .

- 4: D(*) – REAL (KIND=nag_wp) array Input/Output
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: must contain the n diagonal elements of the matrix A .
On exit: if $\text{IFAIL} \geq 0$, D is overwritten by the n diagonal elements of the upper triangular matrix U from the LU factorization of A .
- 5: DU(*) – REAL (KIND=nag_wp) array Input/Output
Note: the dimension of the array DU must be at least $\max(1, N - 1)$.
On entry: must contain the $(n - 1)$ superdiagonal elements of the matrix A
On exit: if $\text{IFAIL} \geq 0$, DU is overwritten by the $(n - 1)$ elements of the first superdiagonal of U .
- 6: DU2(N - 2) – REAL (KIND=nag_wp) array Output
On exit: if $\text{IFAIL} \geq 0$, DU2 returns the $(n - 2)$ elements of the second superdiagonal of U .
- 7: IPIV(N) – INTEGER array Output
On exit: if $\text{IFAIL} \geq 0$, the pivot indices that define the permutation matrix P ; at the i th step row i of the matrix was interchanged with row $\text{IPIV}(i)$. $\text{IPIV}(i)$ will always be either i or $(i + 1)$; $\text{IPIV}(i) = i$ indicates a row interchange was not required.
- 8: B(LDB,*) – REAL (KIND=nag_wp) array Input/Output
Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.
On entry: the n by r matrix of right-hand sides B .
On exit: if $\text{IFAIL} = 0$ or $N + 1$, the n by r solution matrix X .
- 9: LDB – INTEGER Input
On entry: the first dimension of the array B as declared in the (sub)program from which F04BCF is called.
Constraint: $\text{LDB} \geq \max(1, N)$.
- 10: RCOND – REAL (KIND=nag_wp) Output
On exit: if no constraints are violated, an estimate of the reciprocal of the condition number of the matrix A , computed as $\text{RCOND} = 1 / (\|A\|_1 \|A^{-1}\|_1)$.
- 11: ERRBND – REAL (KIND=nag_wp) Output
On exit: if $\text{IFAIL} = 0$ or $N + 1$, an estimate of the forward error bound for a computed solution \hat{x} , such that $\|\hat{x} - x\|_1 / \|x\|_1 \leq \text{ERRBND}$, where \hat{x} is a column of the computed solution returned in the array B and x is the corresponding column of the exact solution X . If RCOND is less than **machine precision**, then ERRBND is returned as unity.
- 12: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by $X04AAF$).

Errors or warnings detected by the routine:

$IFAIL < 0$ and $IFAIL \neq -999$

If $IFAIL = -i$, the i th argument had an illegal value.

$IFAIL = -999$

Allocation of memory failed. The integer allocatable memory required is N , and the real allocatable memory required is $2 \times N$. In this case the factorization and the solution X have been computed, but $RCOND$ and $ERRBND$ have not been computed.

$IFAIL > 0$ and $IFAIL \leq N$

If $IFAIL = i$, u_{ii} is exactly zero. The factorization has been completed, but the factor U is exactly singular, so the solution could not be computed.

$IFAIL = N + 1$

$RCOND$ is less than *machine precision*, so that the matrix A is numerically singular. A solution to the equations $AX = B$ has nevertheless been computed.

7 Accuracy

The computed solution for a single right-hand side, \hat{x} , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. F04BCF uses the approximation $\|E\|_1 = \epsilon \|A\|_1$ to estimate $ERRBND$. See Section 4.4 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating point operations required to solve the equations $AX = B$ is proportional to nr . The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The complex analogue of F04BCF is F04CCF.

9 Example

This example solves the equations

$$AX = B,$$

where A is the tridiagonal matrix

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 2.7 & 6.6 \\ -0.5 & 10.8 \\ 2.6 & -3.2 \\ 0.6 & -11.2 \\ 2.7 & 19.1 \end{pmatrix}.$$

An estimate of the condition number of A and an approximate error bound for the computed solutions are also printed.

9.1 Program Text

Program f04bcfe

```

!      F04BCF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: f04bcf, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: errbnd, rcond
Integer                     :: i, ierr, ifail, ldb, n, nrhs
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: b(:,,:), d(:), dl(:), du(:), du2(:)
Integer, Allocatable        :: ipiv(:)
!      .. Executable Statements ..
Write (nout,*) 'F04BCF Example Program Results'
Write (nout,*)
Flush (nout)
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs
ldb = n
Allocate (b(ldb,nrhs),d(n),dl(n-1),du(n-1),du2(n-2),ipiv(n))
!      Read A and B from data file

Read (nin,*) du(1:n-1)
Read (nin,*) d(1:n)
Read (nin,*) dl(1:n-1)
Read (nin,*) (b(i,1:nrhs),i=1,n)

!      Solve the equations AX = B for X

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 1
Call f04bcf(n,nrhs,dl,d,du,du2,ipiv,b,ldb,rcond,errbnd,ifail)

If (ifail==0) Then
!      Print solution, estimate of condition number and approximate
!      error bound

ierr = 0
Call x04caf('General',' ',n,nrhs,b,ldb,'Solution',ierr)

Write (nout,*)
Write (nout,*) 'Estimate of condition number'
Write (nout,99999) 1.0E0_nag_wp/rcond
Write (nout,*)
Write (nout,*) 'Estimate of error bound for computed solutions'
Write (nout,99999) errbnd
Else If (ifail==n+1) Then
!      Matrix A is numerically singular. Print estimate of

```

```

!      reciprocal of condition number and solution

      Write (nout,*)
      Write (nout,*) 'Estimate of reciprocal of condition number'
      Write (nout,99999) rcond

      Write (nout,*)
      Flush (nout)
      ierr = 0
      Call x04caf('General',' ',n,nrhs,b,ldb,'Solution',ierr)

Else If (ifail>0 .And. ifail<=n) Then
!      The upper triangular matrix U is exactly singular.  Print
!      details of factorization

      Write (nout,*) 'Details of factorization'
      Write (nout,*)
      Write (nout,*) ' Second super-diagonal of U'
      Write (nout,99998) du2(1:n-2)
      Write (nout,*)
      Write (nout,*) ' First super-diagonal of U'
      Write (nout,99998) du(1:n-1)
      Write (nout,*)
      Write (nout,*) ' Main diagonal of U'
      Write (nout,99998) d(1:n)
      Write (nout,*)
      Write (nout,*) ' Multipliers'
      Write (nout,99998) dl(1:n-1)
      Write (nout,*)
      Write (nout,*) ' Vector of interchanges'
      Write (nout,99997) ipiv(1:n)
Else
      Write (nout,99996) ifail
End If

99999 Format (6X,1P,E9.1)
99998 Format (1X,8F9.4)
99997 Format (1X,8I9)
99996 Format (1X,' ** F04BCF returned with IFAIL = ',I5)
      End Program f04bcfe

```

9.2 Program Data

F04BCF Example Program Data

```

5      2      : n, nrhs

      2.1  -1.0  1.9  8.0 : superdiagonal du
3.0    2.3  -5.0  -0.9  7.1 : diagonal d
3.4    3.6   7.0  -6.0      : subdiagonal dl

2.7    6.6
-0.5   10.8
2.6    -3.2
0.6   -11.2
2.7    19.1      : matrix B

```

9.3 Program Results

F04BCF Example Program Results

```

Solution
      1      2
1     -4.0000  5.0000
2      7.0000 -4.0000
3      3.0000 -3.0000
4     -4.0000 -2.0000
5     -3.0000  1.0000

```

Estimate of condition number
9.3E+01

Estimate of error bound for computed solutions
1.0E-14
